

Documented ICBM recalibration equations

In this document I describe the JAGS model file used in the ICBM recalibration.

As a doublecheck, the code is shown here. This is exactly the file currently used by the model (the following code chunk is produced with a relative link). **Please consider this just for documentation, and skip to the end of it, where the code is described in detail with mathematical notation.**

```
1 model{
2
3   ####Loop for treatments
4   for(j in 1:J_Ultuna)
5   {
6
7     Y_R_Ultuna[j,1]<-(SOC_init_Ultuna[j]*(1-Init_ratio_Ultuna))*0.5
8     Y_S_Ultuna[j,1]<-(SOC_init_Ultuna[j]*(1-Init_ratio_Ultuna))*0.5
9     Y_FYM_Ultuna[j,1]<-0
10    Y_GM_Ultuna[j,1]<-0
11    Y_PEA_Ultuna[j,1]<-0
12    Y_SAW_Ultuna[j,1]<-0
13    Y_SLU_Ultuna[j,1]<-0
14    Y_STR_Ultuna[j,1]<-0
15    O_Ultuna[j,1] <-SOC_init_Ultuna[j]*Init_ratio_Ultuna
16
17    # loop for years
18    for (i in 1:(N_Ultuna)){
19
20      alpha_1[j,i]=ifelse(j==1, 0, alpha) #ifelse to have the intercept zero in case of bare fallow
21      alpha_maize_1[j,i]=ifelse(j==1, 0, alpha_maize) #ifelse to have the intercept zero in case of bare fallow
22
23
24      #Inputs R (roots), with different allometric functions for crops
25      #depth coefficients from: Fan, J., McConkey, B., Wang, H., & Janzen, H. (2016). Root distribution by depth for
26      #temperate agricultural crops.
```

```

27 #Field Crops Research, 189, 68-74. https://doi.org/10.1016/j.fcr.2016.02.013
28 I_R_cereals_Ultuna[j,i] <- (1+exudates_coeff)*e_depth_cer*0.5560437*C_percent*
29 (alpha_1[j,i]+Yields_cereals_Ultuna[j,i])*(1/(SR_cereals_ult))
30 I_R_root_crops_Ultuna[j,i] <- (1+exudates_coeff)*e_depth_root*0.5902446*0.32*C_percent*
31 (alpha_1[j,i]+Yields_root_crops_Ultuna[j,i])*(1/(SR_root_crops_ult))#SR_root_crops_ult)
32 I_R_oilseeds_Ultuna[j,i] <- (1+exudates_coeff)*e_depth_oil*0.6367459*C_percent*
33 (alpha_1[j,i]+Yields_oilseeds_Ultuna[j,i])*(1/(SR_oilseeds_ult))#SR_oilseeds_ult)
34 I_R_maize_Ultuna[j,i] <- (1+exudates_coeff)*e_depth_maize*0.5981638*C_percent*
35 (alpha_maize_1[j,i]+Yields_maize_Ultuna[j,i])*(1/(SR_maize_ult))#SR_maize_ult)
36 I_R_Ultuna[j,i] <- I_R_cereals_Ultuna[j,i]+I_R_root_crops_Ultuna[j,i]+I_R_oilseeds_Ultuna[j,i]+I_R_maize_Ultuna[j,i]
37 #Inputs S
38 I_S_Ultuna[j,i] <- ((Yields_cereals_Ultuna[j,i]+Yields_root_crops_Ultuna[j,i]+Yields_oilseeds_Ultuna[j,i])*
39 stubbles_ratio_Ultuna+Yields_maize_Ultuna[j,i]*stubbles_ratio_Ultuna_maize)*C_percent
40
41 #Young R
42 Y_R_Ultuna[j,i+1] <- (I_R_Ultuna[j,i]+Y_R_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
43 Y_S_Ultuna[j,i+1] <- (I_S_Ultuna[j,i]+Y_S_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
44
45 Y_FYM_Ultuna[j,i+1] <- (I_FYM_Ultuna[j,i]+Y_FYM_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
46 Y_GM_Ultuna[j,i+1] <- (I_GM_Ultuna[j,i]+Y_GM_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
47 Y_PEA_Ultuna[j,i+1] <- (I_PEA_Ultuna[j,i]+Y_PEA_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
48 Y_SAW_Ultuna[j,i+1] <- (I_SAW_Ultuna[j,i]+Y_SAW_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
49 Y_SLU_Ultuna[j,i+1] <- (I_SLU_Ultuna[j,i]+Y_SLU_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
50 Y_STR_Ultuna[j,i+1] <- (I_STR_Ultuna[j,i]+Y_STR_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
51
52 #Old
53
54 #old flux
55 fluxR_Ultuna[j,i] <- h_R_ult*((k1_ult*(Y_R_Ultuna[j,i]+I_R_Ultuna[j,i]))/(k2_ult-k1_ult))
56 fluxS_Ultuna[j,i] <- h_S_ult*((k1_ult*(Y_S_Ultuna[j,i]+I_S_Ultuna[j,i]))/(k2_ult-k1_ult))
57
58 flux_FYM_Ultuna[j,i] <- h_FYM_ult*((k1_ult*(Y_FYM_Ultuna[j,i]+I_FYM_Ultuna[j,i]))/(k2_ult-k1_ult))
59 flux_GM_Ultuna[j,i] <- h_S_ult*((k1_ult*(Y_GM_Ultuna[j,i]+I_GM_Ultuna[j,i]))/(k2_ult-k1_ult))
60 flux_PEA_Ultuna[j,i] <- h_PEA_ult*((k1_ult*(Y_PEA_Ultuna[j,i]+I_PEA_Ultuna[j,i]))/(k2_ult-k1_ult))
61 flux_SAW_Ultuna[j,i] <- h_SAW_ult*((k1_ult*(Y_SAW_Ultuna[j,i]+I_SAW_Ultuna[j,i]))/(k2_ult-k1_ult))
62 flux_SLU_Ultuna[j,i] <- h_SLU_ult*((k1_ult*(Y_SLU_Ultuna[j,i]+I_SLU_Ultuna[j,i]))/(k2_ult-k1_ult))
63 flux_STR_Ultuna[j,i] <- h_S_ult*((k1_ult*(Y_STR_Ultuna[j,i]+I_STR_Ultuna[j,i]))/(k2_ult-k1_ult))

```

```

64
65 flux_sum_Ultuna[j,i]<-(fluxR_Ultuna[j,i]+
66     fluxS_Ultuna[j,i]+
67     flux_FYM_Ultuna[j,i]+
68     flux_GM_Ultuna[j,i]+
69     flux_PEA_Ultuna[j,i]+
70     flux_SAW_Ultuna[j,i]+
71     flux_SLU_Ultuna[j,i]+
72     flux_STR_Ultuna[j,i])
73
74
75 O_Ultuna[j,i+1]      <-  (O_Ultuna[j,i]-flux_sum_Ultuna[j,i])*exp(-k2_ult*re_Ultuna[j,i]) +
76     flux_sum_Ultuna[j,i]*exp(-k1_ult*re_Ultuna[j,i])
77
78
79 #Total C
80 Y_tot_Ultuna[j,i] <- Y_R_Ultuna[j,i] +
81     Y_S_Ultuna[j,i] +
82     Y_FYM_Ultuna[j,i] +
83     Y_GM_Ultuna[j,i] +
84     Y_PEA_Ultuna[j,i] +
85     Y_SAW_Ultuna[j,i] +
86     Y_SLU_Ultuna[j,i] +
87     Y_STR_Ultuna[j,i]
88
89 Tot_Ultuna[j,i] <- Y_tot_Ultuna[j,i] + O_Ultuna[j,i]
90
91 #Error of the measurement (assumed proportional to the measurement)
92 SOC_Ultuna[j,i] ~ dnorm(Tot_Ultuna[j,i],1/(error_SOC_Ultuna[j]*1.5))
93 #SOC_Ultuna[j,i] ~ dunif(Tot_Ultuna[j,i]-Tot_Ultuna[j,i]*error_SOC_Ultuna[j],Tot_Ultuna[j,i]+Tot_Ultuna[j,i]*error_SOC_Ultuna[j])
94
95 }
96
97 }
98
99
100

```

```

101  ##Parameters Ultuna
102  # Xie, Yajun. 2020. "A Meta-Analysis of Critique of Litterbag Method Used in Examining Decomposition of Leaf Litters.
103  #" Journal of Soils and Sediments 20 (4): 1881-86. https://doi.org/10.1007/s11368-020-02572-9.
104
105  k1_ult ~ dunif(0.4, 1)
106  k2_ult ~ dunif(0,0.03)
107
108
109  h_S_ult ~ dunif(0.125-0.125*limits_h,0.125+0.125*limits_h)
110  h_R_ult ~ dunif(0.35-0.35*limits_h, 0.35+0.35*limits_h)
111  h_FYM_ult ~ dunif(0.27-0.27*limits_h,0.27+0.27*limits_h)
112  h_PEA_ult ~ dunif(0.59-0.59*limits_h, 0.59+0.59*limits_h)
113  h_SAW_ult ~ dunif(0.25-0.25*limits_h,0.25+0.25*limits_h)
114  h_SLU_ult ~ dunif(0.41-0.41*limits_h,0.41+0.41*limits_h)
115
116  #root/shoot ratios priors
117  SR_cereals_ult ~ dunif(3.6,27.9) #range from martin data
118  #SR_cereals_ult ~ dnorm(11,1/2)
119  SR_root_crops_ult ~ dunif(29.49853-29.49853*limit_SR,29.49853+29.49853*limit_SR)
120  SR_oilseeds_ult ~ dunif(8-8*limit_SR,8+8*limit_SR)
121  SR_maize_ult ~ dunif(6.25-6.25*limit_SR,6.25+6.25*limit_SR)
122
123  #alpha ~dunif(0,1)
124  #alpha_maize ~dunif(0,1)
125  alpha = 0
126  alpha_maize= 0
127
128
129  exudates_coeff ~ dnorm(1.65, 1/0.4125) #10% error
130  e_depth_cer ~ dnorm(1, 1/0.25) #10% error
131  e_depth_root ~ dnorm(1, 1/0.25) #10% error
132  e_depth_oil ~ dnorm(1, 1/0.25) #10% error
133  e_depth_maize ~ dnorm(1, 1/0.25) #10% error
134
135  #Init_ratio_Ultuna ~ dnorm(0.9291667,1/(0.9291667*0.2)) T(0.8,0.98)
136  Init_ratio_Ultuna ~ dunif(0.8,0.98)
137

```

```

138 stubbles_ratio_Ultuna_maize ~ dunif(0.01,0.08)
139 stubbles_ratio_Ultuna ~ dunif(0.01,0.08)
140
141 C_percent ~ dunif(0.40, 0.51)
142
143 limits_h<-1
144 limits_k<-1
145 limit_SR<-1
146
147 }

```

The model starts with a loop for each treatment

```
for(j in 1:J_Ultuna){
```

Where the index “j” is updated for each of the treatments considered in that calibration. Inside this loop there is another loop, nested, that runs for each simulation year “n”

```
for (i in 1:(N_Ultuna)){
```

Inputs (lines 24 to 38)

The two ifelses

```

alpha_1[j,i]=ifelse(j==1, 0, alpha)
alpha_maize_1[j,i]=ifelse(j==1, 0, alpha_maize)

```

are for not using the intercept α when considering the bare fallow. Please note that α is in some calibrations anyway set to zero.

Root inputs are calculated for each year and for each crop type (line 28 to 36). Since the input matrix has zero for each crop that was not that year crop, the sum correspond to the crop of that year.

$$I_{R(c,t)} = (1 + \rho) \cdot \epsilon_d \cdot z \cdot \theta \cdot (\alpha \cdot A_{cr} \cdot \frac{1}{S : R_s}) \quad (1)$$

Where: ρ =exudates coefficient ϵ_d = error term (“e_depth_” in the code) z_{cr} = the fraction of roots at 20 cm depth (Fan et al., 2016) θ = C concentration α = the intercept (in some calibrations it is set to zero) A_{cr} = aboveground biomass $S : R$ = Shoot to root ratio cr =crop t = year We then sum all the crops for each year

$$I_{R(t)} = \sum_{1 \dots s} I_{R(s,t)} \quad (2)$$

With the code:

```
I_R_Ultuna[j,i] <- I_R_cereals_Ultuna[j,i]+I_R_root_crops_Ultuna[j,i]+I_R_oilseeds_Ultuna[j,i]+I_R_maize_Ultuna[j,i]
```

Shoot inputs are calculated with the stubbles ratio ξ :

$$I_{S(s,t)} = A_s \cdot \xi \cdot \theta \quad (3)$$

With the code:

```
((Yields_cereals_Ultuna[j,i]+Yields_root_crops_Ultuna[j,i]+Yields_oilseeds_Ultuna[j,i])*  
stubbles_ratio_Ultuna+Yields_maize_Ultuna[j,i]*stubbles_ratio_Ultuna_maize)*C_percent
```

Amdendment inputs ($I_{n(t)}$) are taken directly as mass of C. Each input is then brought forward in a specific Young pool

Young pool

Each material (roots, shoots, different amendements, all denoted by m) decays into more fine organic matter during the year:

$$Y_{m(t+1)} = (I_{m(t)} + Y_{n(t)}) \cdot e^{-k \cdot r_e} \quad (4)$$

This is repeated for roots, shoots and each of the amendements. In each of the treatments we calculate all these pools, which are just equal to zero in treatments with no inputs in that class:

```
Y_R_Ultuna[j,i+1] <- (I_R_Ultuna[j,i]+Y_R_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
Y_S_Ultuna[j,i+1] <- (I_S_Ultuna[j,i]+Y_S_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
  
Y_FYM_Ultuna[j,i+1] <- (I_FYM_Ultuna[j,i]+Y_FYM_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
Y_GM_Ultuna[j,i+1] <- (I_GM_Ultuna[j,i]+Y_GM_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
Y_PEA_Ultuna[j,i+1] <- (I_PEA_Ultuna[j,i]+Y_PEA_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
Y_SAW_Ultuna[j,i+1] <- (I_SAW_Ultuna[j,i]+Y_SAW_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
Y_SLU_Ultuna[j,i+1] <- (I_SLU_Ultuna[j,i]+Y_SLU_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])  
Y_STR_Ultuna[j,i+1] <- (I_STR_Ultuna[j,i]+Y_STR_Ultuna[j,i])*exp(-k1_ult*re_Ultuna[j,i])
```

Fluxes

Each year we also have a flux from each Young pool to the Old (humified) pool:

$$\Phi_{n(t)} = h_m \cdot \frac{k_1 \cdot (I_{m(t)} + Y_{m(t)})}{k_2 - k_1} \quad (5)$$

This is also repeated for roots, shoots and each of the amendments. In each of the treatments we calculate also all these fluxes, which are just equal to zero in treatments with no inputs in that class:

```
fluxR_Ultuna[j,i]      <- h_R_ult*((k1_ult*(Y_R_Ultuna[j,i]+I_R_Ultuna[j,i]))/(k2_ult-k1_ult))
fluxS_Ultuna[j,i]      <- h_S_ult*((k1_ult*(Y_S_Ultuna[j,i]+I_S_Ultuna[j,i]))/(k2_ult-k1_ult))

flux_FYM_Ultuna[j,i]   <- h_FYM_ult *((k1_ult*(Y_FYM_Ultuna[j,i]+I_FYM_Ultuna[j,i]))/(k2_ult-k1_ult))
flux_GM_Ultuna[j,i]    <- h_S_ult  *((k1_ult*(Y_GM_Ultuna[j,i] +I_GM_Ultuna[j,i]))/(k2_ult-k1_ult))
flux_PEA_Ultuna[j,i]   <- h_PEA_ult *((k1_ult*(Y_PEA_Ultuna[j,i]+I_PEA_Ultuna[j,i]))/(k2_ult-k1_ult))
flux_SAW_Ultuna[j,i]   <- h_SAW_ult *((k1_ult*(Y_SAW_Ultuna[j,i]+I_SAW_Ultuna[j,i]))/(k2_ult-k1_ult))
flux_SLU_Ultuna[j,i]   <- h_SLU_ult *((k1_ult*(Y_SLU_Ultuna[j,i]+I_SLU_Ultuna[j,i]))/(k2_ult-k1_ult))
flux_STR_Ultuna[j,i]   <- h_S_ult  *((k1_ult*(Y_STR_Ultuna[j,i]+I_STR_Ultuna[j,i]))/(k2_ult-k1_ult))
```

Note that, while the Y pool is calculated each year for the following year, the fluxes ϕ are calculated for the current year.

Fluxes are summed all together as

$$\Phi = \sum_{1...m} (\phi_{m(t)}) \quad (6)$$

Which translates in the code as:

```
flux_sum_Ultuna[j,i]<-(fluxR_Ultuna[j,i]+
  fluxS_Ultuna[j,i]+
  flux_FYM_Ultuna[j,i]+
  flux_GM_Ultuna[j,i]+
  flux_PEA_Ultuna[j,i]+
  flux_SAW_Ultuna[j,i]+
  flux_SLU_Ultuna[j,i]+
  flux_STR_Ultuna[j,i])
```

Humified pool

All the fluxes are used to calculate the Old pool:

$$O_{(t+1)} = (O_{(t)} - \Phi) \cdot e^{-k_2 r_e} + \Phi \cdot e^{-k_1 r_e} \quad (7)$$

Which corresponds in the code to:

```
O_Ultuna[j,i+1]      <- (O_Ultuna[j,i]-flux_sum_Ultuna[j,i])*exp(-k2_ult*re_Ultuna[j,i]) +
  flux_sum_Ultuna[j,i]*exp(-k1_ult*re_Ultuna[j,i])
```

Total C

Total C is calculated by summing up together all the Young pools with the Old pool at each time step:

$$SOC_{(t)} = \sum_{1...n} (Y_{n(t)}) + O_{(t)} \quad (8)$$

In the code this is achieved by first summing up together the Young pools and then adding the old pool

```
#Total C
Y_tot_Ultuna[j,i] <- Y_R_Ultuna[j,i] +
                    Y_S_Ultuna[j,i] +
                    Y_FYM_Ultuna[j,i] +
                    Y_GM_Ultuna[j,i] +
                    Y_PEA_Ultuna[j,i] +
                    Y_SAW_Ultuna[j,i] +
                    Y_SLU_Ultuna[j,i] +
                    Y_STR_Ultuna[j,i]

Tot_Ultuna[j,i] <- Y_tot_Ultuna[j,i] + O_Ultuna[j,i]
```

Comparing with the results

The simulation (“Tot_Ultuna”) is then compared with the measured values (“SOC_Ultuna”). Of course this is done each loop step for the ith (year) and jth (treatment) positions. The error in each point is estimated as the error of each measured time series (so the error of a linear regression on that specific time series), increasef by 50% for additional safety.

$$SOC_{measured} \sim \mathcal{N}(SOC_{simulated}, 1.5\sigma_{SOC_{measured}}) \quad (9)$$

In the code this is rendered as:

```
SOC_Ultuna[j,i] ~ dnorm(Tot_Ultuna[j,i], 1/(error_SOC_Ultuna[j]*1.5))
```

Please notice that JAGS uses precision and not error for the normal distribution, hence the error is expressed as $\frac{1}{\sigma}$. The distribution is here assumed as gaussian (maybe uniform would be more conservative)

The priors

Outside the two loops, after line 97, all priors are specified. The JAGS syntax is pretty intuitive in this sense. The uniform distribution has as parameters just minimum and maximum:

```
param ~ dunif(min, max)
```

While the normal distribution:

```
param ~ dnorm(mean, precision)
```